6. přednáška

Diferenciální rovnice

Základním příkazem pro řešení diferenciálních rovnic v Maplu je příkaz dsolve.

Diferenciální rovnice 1. řádu

```
> restart;
```

Příklad:

Řešte rovnici $y'=x^2$.

```
> dsolve(diff(y(x),x)=x^2,y(x));
```

$$y(x) = \frac{x^3}{3} + \_C1$$

Vidíme, že naši rovnici splňuje nekonečně mnoho funkcí.

Jiná možnost zápisu:

```
> dsolve(D(y)(x)=x^2,y(x));
```

$$y(x) = \frac{x^3}{3} + \_C1$$

```
>
```

Příklad:

Najděte obecné řešení rovnice $y'=y$.

```
> dsolve(D(y)(x)=y(x),y(x));
```

$$y(x) = \_C1\, \mathbf{e}^x$$

Opět jsme dostali nekonečně mnoho řešení.

Pokud bychom chtěli, aby řešení bylo určeno jednoznačně, museli bychom kromě rovnice zadat ještě tzv. počáteční podmínku / podmínky. Rovnici společně s takovou počáteční podmínkou budeme nazývat počáteční (Cauchyovou) úlohou. Za vhodných předpokladů má Cauchyova úloha právě jedno řešení.

```
> dsolve({D(y)(x)=y(x),y(0)=1},y(x));
```

$$y(x) = e^x$$

Řešení už je určeno jednoznačně.

Graf řešení:

```
> reseni:=rhs(%);
```

$$reseni := e^x$$

rhs ... right hand side, podobně lhs ... left hand side

```
> plot(reseni,x=-2..2,thickness=5);
```




```
>
```

Příklad:

```
> rovnice:=D(y)(x)=(sin(x)+y(x)^2)/x;
```

$$rovnice := D(y)(x) = \frac{\sin(x) + y(x)^2}{x}$$

```
> dsolve({rovnice,y(1)=0},y(x));
```

Vidíme, že Maple nenašel žádné řešení výše uvedené Cauchyovy úlohy. Z teorie ale vyplývá, že řešení takové Cauchyovy úlohy existuje.
Můžeme zkusit řešení spočítat numericky.

```
> numericky:=dsolve({rovnice,y(1)=0},y(x),numeric):
```

```
> numericky(1);
```

$$[x = 1., y(x) = 0.]$$

```
> numericky(2);
```

$$[x = 2., y(x) = 0.779424276500193]$$

Nyní lze použít příkazu odeplot z balíku plots a alespoň nakreslit graf řešení.

```
> plots[odeplot](numericky,x=0.15..5,thickness=5);
```



K předchozímu příkladu se ještě jednou vrátíme (v kapitole o směrových polích).

```
>
```

Příklad:

```
> rovnice:=D(y)(x)=2*sqrt(y(x));
```

$$rovnice := \mathrm{D}(y)(x) = 2\sqrt{\mathrm{y}(x)}$$

```
> dsolve({rovnice,y(1)=1},y(x));
```

$$\mathrm{y}(x) = x^2$$

Zamyslete se nad výsledkem !!!

```
>
```

Nyní si ukážeme některé úlohy vedoucí na diferenciální rovnice prvního řádu.

Příklad (radioaktivní rozpad):

Nechť $\mathrm{y}(t)$ značí množství radioaktivní látky v čase t. Pak y splňuje diferenciální rovnici $y' = -c\,y$. Konstanta c přitom souvisí s radioaktivní látkou a s jejím poločasem rozpadu.

```
> restart;
> rovnice:=D(y)(t)=-c*y(t);
```

$$rovnice := \mathrm{D}(y)(t) = -c\,\mathrm{y}(t)$$

```
> dsolve({rovnice,y(0)=p},y(t));
```

$$y(t) = p\,\mathbf{e}^{(-c\,t)}$$

p je počáteční množství radioaktivní látky.

```
> reseni:=rhs(%);
```

$$reseni := p\,\mathbf{e}^{(-c\,t)}$$

Zvolme si konkrétní hodnoty:

```
> p:=1;
```

$$p := 1$$

```
> c:=0.1;
```

$$c := 0.1$$

Graf řešení:

```
> plot(reseni,t=0..50,thickness=5);
```




```
>
```

Příklad (růst lidské populace - naivní model):

Nechť $y(t)$ představuje počet členů populace v čase t. Pak y splňuje diferenciální rovnici $y' = k\,y$, kde konstanta k charakterizuje populaci a její schopnost množit se.

```
> restart;
> dsolve({D(y)(t)=k*y(t),y(0)=p},y(t));
```

$$y(t) = p\,\mathbf{e}^{(k\,t)}$$

p je počáteční stav populace.

```
> reseni:=rhs(%);
```

$$reseni := p\, \mathbf{e}^{(k\, t)}$$

Zvolme si konkrétní hodnoty:

```
> p:=1000;
```

$$p := 1000$$

```
> k:=0.1;
```

$$k := 0.1$$

Graf řešení:

```
> plot(reseni,t=0..20,thickness=5);
```



Dostali jsme exponenciální růst populace.

Tento model má své nedostatky (např. omezené zásoby potravy, apod.).

```
>
```

Příklad (model růstu populace - upravený):

```
> restart;
> rovnice:=D(y)(t)=k*y(t)*(l-y(t));
```

$$rovnice := \mathrm{D}(y)(t) = k\, \mathrm{y}(t)\, (l - \mathrm{y}(t))$$

```
> dsolve({rovnice,y(0)=p},y(t));
```

$$\mathrm{y}(t) = \frac{p\, l}{p + \mathbf{e}^{(-l\, k\, t)}\, l - \mathbf{e}^{(-l\, k\, t)}\, p}$$

```
> reseni:=rhs(%);
```

$$reseni := \frac{p\, l}{p + \mathbf{e}^{(-l\, k\, t)}\, l - \mathbf{e}^{(-l\, k\, t)}\, p}$$

Konkrétní situace:

```
> p:=1000;  # Počáteční stav
```

$$p := 1000$$

```
> k:=10^(-6);
```

$$k := \frac{1}{1000000}$$

```
> l:=10000; # Maximální dlouhodobě udržitelný počet členů
  populace
```

$$l := 10000$$

Graf řešení:

```
> reseni;
```

$$\frac{10000000}{1000 + 9000\,\mathbf{e}^{\left(-\frac{t}{100}\right)}}$$

```
> plot([reseni,l/2,l],t=0..700,thickness=[5,1,1],linestyle=[
  solid,dot,solid],color=[red,black,blue]);
```



Počet členů populace se limitně blíží k hodnotě l (v našem případě 10 000) - tento model už je realističtější.

Jiný počáteční stav ($p > l$):

```
> p:=15000;
```

$$p := 15000$$

```
> plot(reseni,t=0..700,thickness=5);
```



Počáteční stav byl větší, než je udržitelná mez, proto populace vymírá a počet se postupně stabilizuje u hodnoty l.

>

Příklad (šnek na gumě):

Uvažujme dokonale pružnou gumu délky g, která je připevněna jedním koncem ke zdi. Na druhý konec položíme šneka a necháme ho lézt směrem ke zdi. Šnek je přitom schopen se pohybovat rychlostí v. Zároveň druhý konec gumy chytneme a natahujeme ho konstantní rychlostí w směrem od zdi. Určete, zda (za jakých předpokladů) šnek doleze ke zdi. V kladném případě také určete čas, za který to šnek zvládne.

Řešení:

Nechť $s(t)$ představuje vzdálenost šneka od zdi v čase t.
Úloha pak vede na následující Cauchyovu úlohu:

```
> restart;
> rovnice:=D(s)(t)=s(t)*w/(g+w*t)-v;
```

$$rovnice := D(s)(t) = \frac{s(t)\, w}{g + w\, t} - v$$

```
> poc_podminka:=s(0)=g;
```

$$poc\_podminka := s(0) = g$$

```
> dsolve({rovnice,poc_podminka},s(t));
```

$$s(t) = \left(-\frac{v \ln(g + w\, t)}{w} + \frac{w + v \ln(g)}{w}\right)(g + w\, t)$$

Dostali jsme závislost vzdálenosti šneka od zdi na čase. Šnek ke zdi dorazí, právě když má rovnice $s(t) = 0$ kladný kořen. Hodnota tohoto kořene představuje čas, za který šnek doleze ke zdi.

```
> zavislost:=rhs(%);
```

$$zavislost := \left(-\frac{v\ln(g+w\,t)}{w}+\frac{w+v\ln(g)}{w}\right)(g+w\,t)$$

```
> cas:=solve(zavislost,t);
```

$$cas := \frac{\mathbf{e}^{\left(\frac{w+v\ln(g)}{v}\right)}-g}{w}$$

Upravíme:

```
> cas:=expand(cas);
```

$$cas := \frac{\mathbf{e}^{\left(\frac{w}{v}\right)}g}{w}-\frac{g}{w}$$

```
> cas:=factor(cas);
```

$$cas := \frac{g\left(\mathbf{e}^{\left(\frac{w}{v}\right)}-1\right)}{w}$$

Vidíme, že nalezený kořen je kladný za všech okolností, tzn., že šnek vždy ke zdi doleze!!! (a to i v případě, že gumu natahujeme mnohonásobně vyšší rychlostí, než je rychlost šneka)

Nyní si zvolme konkrétní hodnoty:

```
> g:=1/5;    # Guma na začátku měří 20 cm.
```

$$g := \frac{1}{5}$$

```
> v:=1/1000;    # Rychlost šneka je 1 mm/s.
```

$$v := \frac{1}{1000}$$

```
> w:=5/1000;    # Gumu napínáme rychlostí 5 mm/s.
```

$$w := \frac{1}{200}$$

```
> zavislost;
```

$$\left(-\frac{1}{5}\ln\left(\frac{1}{5}+\frac{t}{200}\right)+1-\frac{1}{5}\ln(5)\right)\left(\frac{1}{5}+\frac{t}{200}\right)$$

```
> cas;
```

$$40\,\mathbf{e}^{5}-40$$

```
> evalf(cas);
```

$$5896.526364$$

V tomto případě tedy šnek doleze ke zdi zhruba za 5897 sekund (1 hodina 38 minut 17 sekund).

Průběh šnekova lezení:

```
> plot(zavislost,t=0..5897,thickness=5);
```




```
>
```

Příklad (problém tchýně a nadzvukového pilota):

Ve vzdálenosti d od domu, kde bydlí tchýně, je letiště. Tam pracuje její zeť (nadzvukový pilot). Pilot vyrazí z letiště nadzvukovou rychlostí m [Machovo číslo, $m > 1$]. Určete, po jaké dráze má pilot letět, aby veškerý hluk z motorů letadla dorazil k domu tchýně v jeden okamžik.

Řešení:

Popíšeme polohu letadla v polárních souřadnicích, tj. $r = \mathrm{f}(t)$. Po sérii úvah dojdeme k tomu, že funkce f musí splňovat diferenciální rovnici $\mathrm{f}\,'(t) = -\dfrac{\mathrm{f}(t)}{\sqrt{m^2-1}}$ a počáteční podmínku $\mathrm{f}(0) = d$. Vyřešme tedy tuto Cauchyovu úlohu.

```
> restart;
> rovnice:=D(f)(t)=-f(t)/sqrt(m^2-1);
```

$$rovnice := \mathrm{D}(f)(t) = -\frac{\mathrm{f}(t)}{\sqrt{m^2-1}}$$

```
> dsolve({rovnice,f(0)=d},f(t));
```

$$\mathrm{f}(t) = d\,\mathbf{e}^{\left(-\frac{t}{\sqrt{m^2-1}}\right)}$$

```
> reseni:=unapply(rhs(%),d,m,t);
```

$$reseni := (d, m, t) \to d\, \mathbf{e}^{\left(-\frac{t}{\sqrt{m^2-1}}\right)}$$

Vytvoříme si proceduru, která bude kreslit optimální dráhu letu.

```
> with(plots):
> kresli:=proc(d,m) local t,p1,p2,p3;
  p1:=polarplot(reseni(d,m,t),t=0..infinity);
  p2:=pointplot([[0,0],[d,0]],symbol=soliddiamond,symbolsize
  =20,color=[black,blue]):display([p1,p2],axes=normal,thickn
  ess=5);
  end;
```

kresli := **proc**(d, m)
local t, $p1$, $p2$, $p3$;
$p1$:= $plots$:-$polarplot$(reseni(d, m, t), $t = 0 .. \infty$);
$p2$:= $plots$:-$pointplot$([[0, 0], [d, 0]], $symbol = soliddiamond$, $symbolsize = 20$, $color = [black, blue]$);
$plots$:-$display$([$p1$, $p2$], $axes = normal$, $thickness = 5$)
end proc

Zvolme si konkrétní hodnoty:
Rychlost letadla bude 2 Machy a vzdálenost letiště od domu tchýně bude 1.

```
> kresli(1,2);
```



A jak to dopadne pro rychlost letadla 5 Machů?

```
> kresli(1,5);
```



Pilot tedy musí letět po jakési spirále.

```
>
```

Směrová pole

```
> restart;
> with(DEtools);
```

[*AreSimilar, Closure, DEnormal, DEplot, DEplot3d, DEplot_polygon, DFactor, DFactorLCLM, DFactorsols, Dchangevar, Desingularize, FunctionDecomposition, GCRD, Gosper, Heunsols, Homomorphisms, IVPsol, IsHyperexponential, LCLM, MeijerGsols, MultiplicativeDecomposition, ODEInvariants, PDEchangecoords, PolynomialNormalForm, RationalCanonicalForm, ReduceHyperexp, RiemannPsols, Xchange, Xcommutator, Xgauge, Zeilberger, abelsol, adjoint, autonomous, bernoullisol, buildsol, buildsym, canoni, caseplot, casesplit, checkrank, chinisol, clairautsol, constcoeffsols, convertAlg, convertsys, dalembertsol, dcoeffs, de2diffop, dfieldplot, diff_table, diffop2de, dperiodic_sols, dpolyform, dsubs, eigenring, endomorphism_charpoly, equinv, eta_k, eulersols, exactsol, expsols, exterior_power, firint, firtest, formal_sol, gen_exp, generate_ic, genhomosol, gensys, hamilton_eqs, hypergeomsols, hyperode, indicialeq, infgen, initialdata, integrate_sols, intfactor, invariants, kovacicsols, leftdivision, liesol, line_int, linearsol, matrixDE, matrix_riccati, maxdimsystems, moser_reduce, muchange, mult, mutest, newton_polygon, normalG2, ode_int_y, ode_y1, odeadvisor, odepde, parametricsol, particularsol, phaseportrait, poincare, polysols, power_equivalent, rational_equivalent, ratsols, redode, reduceOrder, reduce_order, regular_parts, regularsp, remove_RootOf, riccati_system, riccatisol, rifread, rifsimp, rightdivision, rtaylor, separablesol, singularities, solve_group, super_reduce, symgen, symmetric_power, symmetric_product, symtest, transinv, translate, untranslate, varparam, zoom*]

```
> rovnice:=D(y)(x)=y(x);
```

$$rovnice := D(y)(x) = y(x)$$

```
> DEplot(rovnice,y(x),x=-1..1,y=-1..1);
```



Tento příkaz nám vykreslil směrové pole pro danou rovnici.

```
> DEplot(rovnice,y(x),x=-1..1,y=-1..3,arrows=medium,[y(0)=1]
  ,linecolor=black,thickness=5);
```



Nyní jsme dostali směrové pole společně s řešením splňujícím danou počáteční podmínku.

```
>
```

Příklad (už jsme jej řešili numericky):

```
> rovnice:=D(y)(x)=(sin(x)+y(x)^2)/x;
```

$$rovnice := D(y)(x) = \frac{\sin(x) + y(x)^2}{x}$$

```
> dsolve({rovnice,y(1)=0},y(x));
```

Maple nenašel žádné řešení této Cauchyovy úlohy. Z teorie však vyplývá, že takové řešení existuje.
Pomocí příkazu DEplot najdeme aspoň graf tohoto řešení (společně se směrovým polem).

```
> DEplot(rovnice,y(x),x=0.15..5,y=-5..5,[y(1)=0],arrows=large,linecolor=blue,thickness=5);
```




```
>
```

Diferenciální rovnice vyššího řádu

Příklad:

Najděte obecné řešení rovnice $y'' + y = 0$.

```
> restart;
> rovnice:=D(D(y))(x)+y(x)=0;
```

$$rovnice := (D^{(2)})(y)(x) + y(x) = 0$$

```
> dsolve(rovnice,y(x));
```

$$y(x) = \_C1 \sin(x) + \_C2 \cos(x)$$

Vidíme, že řešení není určeno jednoznačně.

Opět je možné uvažovat počáteční podmínky.

```
> dsolve({rovnice,y(0)=0},y(x));
```

$$y(x) = \_C1 \sin(x)$$

```
> dsolve({rovnice,y(0)=1,D(y)(0)=2},y(x));
```

$$y(x) = 2 \sin(x) + \cos(x)$$

```
>
```

Příklad:

```
> rovnice:=(D@@6)(y)(x)=y(x);
```

$$rovnice := (D^{(6)})(y)(x) = y(x)$$

```
> dsolve(rovnice,y(x));
```

$$y(x) = \_C1\, \mathbf{e}^{x} + \_C2\, \mathbf{e}^{(-x)} + \_C3\, \mathbf{e}^{\left(\frac{x}{2}\right)} \sin\left(\frac{\sqrt{3}\, x}{2}\right) + \_C4\, \mathbf{e}^{\left(\frac{x}{2}\right)} \cos\left(\frac{\sqrt{3}\, x}{2}\right) + \_C5\, \mathbf{e}^{\left(-\frac{x}{2}\right)} \sin\left(\frac{\sqrt{3}\, x}{2}\right) + \_C6\, \mathbf{e}^{\left(-\frac{x}{2}\right)} \cos\left(\frac{\sqrt{3}\, x}{2}\right)$$

```
>
```

Příklad (kmitání závaží na pružině):

Uvažujme pružinu, na kterou zavěsíme závaží. Poté pružinu natáhnene a pustíme. Popište závislost výchylky zavěšeného tělesa na čase.

Řešení:
Nechť $y(t)$ představuje výchylku zavěšeného tělesa v čase t. Pak y splňuje následující diferenciální rovnici (druhého řádu):

```
> restart;
> rovnice:=m*D(D(y))(t)=-s*y(t)-r*D(y)(t)+v;
```

$$rovnice := m\, (D^{(2)})(y)(t) = -s\, y(t) - r\, D(y)(t) + v$$

```
> poc_podminky:=y(0)=-d,D(y)(0)=0;
```

$$poc\_podminky := y(0) = -d, D(y)(0) = 0$$

m - hmotnost zavěšeného tělesa

s - konstanta související s tuhostí pružiny

r - konstanta související s odporem prostředí

d - počáteční natažení pružiny

v - vnější síly působící na těleso

Konkrétní situace (žádné vnější síly, žádný odpor prostředí):

```
> m:=1;
```

$$m := 1$$

```
> s:=1;
```

$$s := 1$$

```
> r:=0;
```

$$r := 0$$

```
> d:=1;
```

$$d := 1$$

```
> v:=0;
```

$$v := 0$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t)$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -\cos(t)$$

```
> reseni:=rhs(%);
```

$$reseni := -\cos(t)$$

```
> plot(reseni,t=0..50,thickness=5);
```



Dostali jsme harmonické kmity.

```
>
```

Jiná volba (nenulový - relativně malý - odpor prostředí, žádné vnější síly):

```
> r:=1/5;
```

$$r := \frac{1}{5}$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t) - \frac{1}{5} D(y)(t)$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -\frac{1}{33}\sqrt{11}\, e^{\left(-\frac{t}{10}\right)} \sin\left(\frac{3\sqrt{11}\,t}{10}\right) - e^{\left(-\frac{t}{10}\right)} \cos\left(\frac{3\sqrt{11}\,t}{10}\right)$$

```
> reseni:=rhs(%);
```

$$reseni := -\frac{1}{33}\sqrt{11}\, e^{\left(-\frac{t}{10}\right)} \sin\left(\frac{3\sqrt{11}\,t}{10}\right) - e^{\left(-\frac{t}{10}\right)} \cos\left(\frac{3\sqrt{11}\,t}{10}\right)$$

```
> plot(reseni,t=0..50,thickness=5);
```



Vidíme, že kmity jsou tlumené.

```
>
```

Jiná volba (nenulový - relativně velký - odpor prostředí, žádné vnější síly):

```
> r:=3;
```

$$r := 3$$

```
> rovnice;
```

$$(\mathrm{D}^{(2)})(y)(t) = -\mathrm{y}(t) - 3\,\mathrm{D}(y)(t)$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := \mathrm{y}(t) = \left(-\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\mathbf{e}^{\left(\frac{(\sqrt{5}-3)\,t}{2}\right)} + \left(-\frac{1}{2} + \frac{3\sqrt{5}}{10}\right)\mathbf{e}^{\left(-\frac{(\sqrt{5}+3)\,t}{2}\right)}$$

```
> reseni:=rhs(%);
```

$$reseni := \left(-\frac{1}{2} - \frac{3\sqrt{5}}{10}\right)\mathbf{e}^{\left(\frac{(\sqrt{5}-3)\,t}{2}\right)} + \left(-\frac{1}{2} + \frac{3\sqrt{5}}{10}\right)\mathbf{e}^{\left(-\frac{(\sqrt{5}+3)\,t}{2}\right)}$$

```
> plot(reseni,t=0..20,thickness=5);
```



V tomto případě vidíme, že nedochází ke kmitům. Vychýlené těleso se pouze vrací do své původní polohy.

```
>
```

Další volba (žádný odpor prostředí, nenulové vnější síly):

```
> r:=0;
```

$$r := 0$$

```
> v:=-1;
```

$$v := -1$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t) - 1$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -1$$

```
> plot(rhs(reseni),t=0..50,thickness=5);
```




```
>
> v:=-1/2;
```

$$v := \frac{-1}{2}$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t) - \frac{1}{2}$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -\frac{1}{2} - \frac{1}{2}\cos(t)$$

```
> plot(rhs(reseni),t=0..50,thickness=5);
```




```
>
> v:=-sin(2*t);
```

$$v := -\sin(2\,t)$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t) - \sin(2\,t)$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -\frac{2}{3}\sin(t) - \cos(t) + \frac{1}{3}\sin(2\,t)$$

```
> plot(rhs(reseni),t=0..50,thickness=5);
```




```
> 
> v:=-sin(t);
```

$$v := -\sin(t)$$

```
> rovnice;
```

$$(D^{(2)})(y)(t) = -y(t) - \sin(t)$$

```
> reseni:=dsolve({rovnice,poc_podminky},y(t));
```

$$reseni := y(t) = -\frac{1}{2}\sin(t) - \cos(t) + \frac{1}{2}\cos(t)\,t$$

```
> plot(rhs(reseni),t=0..50,thickness=5);
```



Zde dochází k rezonanci - po nějakém čase by se pružina utrhla.

Podobně bychom si mohli "hrát" dál.

```
> 
```

Soustavy diferenciálních rovnic

Příklad:

```
> restart;
> soustava:=D(u)(t)=u(t)-v(t),D(v)(t)=u(t)+v(t);
```

$$soustava := D(u)(t) = u(t) - v(t),\ D(v)(t) = u(t) + v(t)$$

```
> dsolve({soustava},{u(t),v(t)});
```

$$\{u(t) = e^t\,(\_C1\,\sin(t) + \_C2\,\cos(t)),\ v(t) = e^t\,(-\_C1\,\cos(t) + \_C2\,\sin(t))\}$$

Opět je možné uvažovat počáteční podmínky.

```
> dsolve({soustava,u(0)=1,v(0)=2},{u(t),v(t)});
```

$$\{u(t) = e^t\,(-2\,\sin(t) + \cos(t)),\ v(t) = e^t\,(2\,\cos(t) + \sin(t))\}$$

Možné grafické znázornění:

```
> reseni:=%;
```

$$reseni := \{u(t) = e^t(-2\sin(t) + \cos(t)), v(t) = e^t(2\cos(t) + \sin(t))\}$$

```
> reseni:=map(x->rhs(x),[op(reseni)]);
```

$$reseni := [e^t(-2\sin(t) + \cos(t)), e^t(2\cos(t) + \sin(t))]$$

```
> plot([op(reseni),t=0..4],thickness=5,labels=[`u(t)`,`v(t)`
  ]);
```



Jiný způsob:

```
> with(DEtools):
> DEplot([soustava],[u(t),v(t)],t=0..4,[[u(0)=1,v(0)=2]],u=-
  50..50,v=-110..10,arrows=medium,color=black,linecolor=red,
  thickness=5);
```




```
>
```

Příklad:

```
> soustava:=D(u)(t)=u(t)+2*v(t)+t,D(v)(t)=2*u(t)+v(t)-exp(t)
;
```

$$soustava := D(u)(t) = u(t) + 2\,v(t) + t,\ D(v)(t) = 2\,u(t) + v(t) - e^{t}$$

```
> dsolve({soustava},{u(t),v(t)});
```

$$\{u(t) = e^{(-t)}\,\_C2 + e^{(3\,t)}\,\_C1 - \frac{5}{9} + \frac{t}{3} + \frac{1}{2}e^{t},\ v(t) = -e^{(-t)}\,\_C2 + e^{(3\,t)}\,\_C1 + \frac{4}{9} - \frac{2\,t}{3}\}$$

```
> dsolve({soustava,u(0)=0,v(0)=0},{u(t),v(t)});
```

$$\{u(t) = \frac{1}{4}e^{(-t)} - \frac{7}{36}e^{(3\,t)} - \frac{5}{9} + \frac{t}{3} + \frac{1}{2}e^{t},\ v(t) = -\frac{1}{4}e^{(-t)} - \frac{7}{36}e^{(3\,t)} + \frac{4}{9} - \frac{2\,t}{3}\}$$

```
>
```

Příklad (model dravec-kořist):

Předpokládejme, že dravec a kořist žijí izolovaně od ostatních druhů a že dravec se živí výhradně kořistí. Kdyby byl dravec izolován od kořisti, pak by vymřel. Dále předpokládejme, že kořist má dostatečný zdroj potravy.

Označme:
$x(t)$ velikost populace kořisti v čase t.
$y(t)$ velikost populace dravce v čase t.

Pak $x(t)$ a $y(t)$ splňují následující systém diferenciálních rovnic:

$x' = (a - b\,y)\,x,$
$y' = (-c + d\,x)\,y.$

a, b, c, d jsou přitom kladné konstanty.

```
> restart;
> with(DEtools):
> rovnice1:=D(x)(t)=(a-b*y(t))*x(t);
```

$$rovnice1 := D(x)(t) = (a - b\,y(t))\,x(t)$$

```
> rovnice2:=D(y)(t)=(-c+d*x(t))*y(t);
```

$$rovnice2 := D(y)(t) = (-c + d\,x(t))\,y(t)$$

Zvolme si konkrétní hodnoty konstant:

```
> a:=1;
```

$$a := 1$$

```
> b:=1;
```

$$b := 1$$

```
> c:=1;
```

$$c := 1$$

```
> d:=1;
```

$$d := 1$$

```
> rovnice1,rovnice2;
```

$$D(x)(t) = (1 - y(t))\,x(t),\ D(y)(t) = (-1 + x(t))\,y(t)$$

```
> dsolve({rovnice1,rovnice2},{x(t),y(t)});
```

$$\left[\{x(t) = 0\},\ \{y(t) = \_C1\,\mathrm{e}^{(-t)}\}\right], \left[\left\{x(t) = \mathrm{RootOf}\left(-\int^{\_Z} \frac{1}{\_a + \mathrm{e}^{\left(-\mathrm{LambertW}\left(\frac{\mathrm{e}^{-\_a}\,\mathrm{e}^{-\_C1}\,\mathrm{e}^{(-1)}}{\_a}\right) + \_a + \_C1 - 1\right)}}\, d\_a + t + \_C2\right)\right\},\ \left\{y(t) = \frac{-\left(\frac{d}{dt}x(t)\right) + x(t)}{x(t)}\right\}\right]$$

```
> DEplot([rovnice1,rovnice2],[x(t),y(t)],t=0..8,x=0..3,y=0..
3,[seq([x(0)=i,y(0)=1],i=0..1,1/5),[x(0)=1,y(0)=0]],arrows
=large,linecolor=black,thickness=5,animatecurves=false);
```




```
>
```

Cvičení

1) Najděte funkci f splňující rovnost $f'(x) = x^2 - \frac{2 f(x)}{x}$ (pro každé kladné x) takovou, jejíž graf prochází bodem $(1, 2)$. Jaká je hodnota této funkce v bodě 3 ?

2) Najděte funkci f definovanou na intervalu $\langle 0, 1\rangle$, jejíž graf leží v prvním kvadrantu a prochází body $(0, 1)$ a $(1, 0)$, s následující vlastností:
Sestrojíme-li v jakémkoliv bodě tečnu ke grafu funkce f, bude součet délek úseků, které vytne tato tečna na osách, roven jedné (viz animace níže).

Najděte předpis pro tuto funkci. Jaká je hodnota $f\left(\frac{1}{2}\right)$?



(Součet délek černých úseček má být vždy roven 1. Animace slouží pouze jako ilustrace - funkce na ní nesplňuje podmínku se součtem 1.)

3) Uvažujme úlohu se šnekem na gumě. Určete **(obecně)** ve kterém čase se šnek nachází nejdále od zdi. Jak velká je ona maximální vzdálenost? Poté uvažujte konkrétní situaci $g = \frac{1}{5}$, $v = \frac{1}{1000}$ a $w = \frac{5}{1000}$ (stejně jako v přednášce) a určete maximální šnekovu vzdálenost od zdi s přesností na milimetry.

4) Uvažujme úlohu o tchýni a pilotovi a předpokládejme, že vzdálenost letiště od domu tchýně je 1 a rychlost letadla je 5 Machů. Z přednášky víme, že pilot musí letět po spirálovité dráze.



Určete, jakou vzdálenost pilot uletí, než se zřítí na dům tchýně.
V jaké vzdálenosti od domu tchýně bude pilot v okamžiku, kdy urazí vzdálenost 3?

>